Simultaneous Modelling of Disease Progression and Time to Event with NONMEM

Nick Holford

Dept of Pharmacology & Clinical Pharmacology, University of Auckland, Private Bag 92019, Auckland, New Zealand

Hu & Sale Terminology

Variables are dropout time *T*, observed (Y_O) and unobserved values (Y_U) of disease progress state (e.g. HIV viral load)

(a) completely random (CRD), if *Ti* is independent of η, and therefore (Y_O, Y_U);

(b) **random** (RD), if *Ti* (given Y_O) is independent of Y_U, but may depend on Y_O. In addition, any dependence of *Ti* on η is only through Y_O;

(c) **informative** (ID), if *Ti* (given Y_O) depends on Y_U, or explicitly depends on η other than through Y_O.

Objective: An important challenge for clinical pharmacologists is to be able to describe the time course of disease progression biomarkers and link this to the probability of clinical outcome events. A common event in clinical trials is subject dropout. Hu & Sale described a joint modeling method for describing informative dropout using observations of a disease status biomarker and a subject dropout interval (the exact time of dropout was not known) or censoring time. They used NM-TRAN to construct code for -2 times the log likelihood (-2LL) for each type of observation. The objective of this study is to compare the NM-TRAN method with using a modified CCONTR subroutine to compute the objective function contributions and to evaluate the use of the likelihood ratio test for model discrimination.

Methods: The -2LL method has been compared with the CCONTR method using NM-TRAN to compute the likelihood for dropout and censoring events and the more usual predicted value for the continuous scale disease status. Biomarker status, dropout and censoring event data were simulated with NONMEM. Data was simulated and parameters estimated using a linear time course for the disease status and 3 dropout models (completely at random, random and informative). NONMEM was used to estimate parameters of the joint model. A randomization test was used to generate null distributions for the likelihood ratio (LR) obtained from data simulated with completely random dropout.
Results: The CCONTR method had more successful runs (79% vs 44%) and was 10% faster (100 runs) than the NM-TRAN method. The estimates of slope and parameter variability of the disease status were unbiased for both methods. The CCONTR method estimates of baseline hazard and informative dropout hazard were also unbiased but the NM-TRAN method estimates were significantly biased (+15% and -2% respectively). The root mean square error of all parameters was less than 20%. The null distribution of the LR obtained from random and informative dropout models fitted to completely random dropout data was similar to the chi-square distribution.
Conclusion: NONMEM can be used to estimate hazard function parameters for dropout models with acceptable bias and imprecision. The CCONTR method is preferable to NM-TRAN coding of -2LL for joint models. Model discrimination can be performed by assuming the likelihood ratio is approximately chi-square distributed.
Reference: Hu C, Sale ME. A joint model for nonlinear longitudinal data with informative dropout. J Pharmacokinet Pharmacodyn 2003;30(1):83-103.

Probability of Event

IF(DV.EQ.0)hasnotdroppedout

$$\Pr_{ID} = e^{-\int_0^t \beta_0 e^{\beta_2 \cdot Disprg(t)}}$$

ELSE; droppedoutbetween $t_{lastobs}$ and t_{end}

$$\Pr_{ID} = e^{-\int_0^{t_{lastobs}} \beta_0 e^{\beta_2 \cdot Disprg(t)}} \cdot \left(1 - e^{-\int_{t_{lastobs}}^{t_{end}} \beta_0 e^{\beta_2 \cdot Disprg(t)}}\right)$$

$_2$ is a parameter describing the informative dropout hazard

Hu & Sale Code

```
$MODEL COMP=CUMHAZ ; compartment for integration of hazard
COMP=(HZLAST, INITIALOFF) ; comp for LAST PERIOD hazard
$PK
INTERC=(THETA(1) - THETA(2)*(TRT-1))+ETA(1)
SLOPE=THETA(3)+ETA(2)
BSHZ=THETA(4)
BETA=THETA(5)
BET2=THETA(6)
$DES
VIRL=INTERC+SLOPE*(T-12)
TEMP=BETA*LOCF+BET2*VIRL
DADT(1)=EXP(TEMP)
DADT(2)=EXP(TEMP)
$ERROR
CMHZ=BSHZ*A(1)
HZLA=BSHZ*A(2)
IF (DVID.EQ.1) THEN ; DV=Viral Load
IPRE=INTERC+SLOPE*(TIME-12)
Y=2*LOG(THETA(7))+( (DV-IPRE)/THETA(7) )**2
ENDIF
IF (DVID.EQ.2 .AND. DV.EQ.0) THEN ; NO dropout
Y=-2*(-CMHZ)
ENDIF
IF (DVID.EQ.2 .AND. DV.EQ.1) THEN ; dropout
Y=-2*(-(CMHZ-HZLA)) - 2*LOG(1 - EXP(-HZLA))
ENDIF
```

-2 Log Likelihood

Disease Progress

$$ELS = \sum_{i=1}^{NSUBNOBS} \sum_{j=1} \left(\frac{\left(Y_{OBSij} - Y_{PREDij}\right)^2}{Var_{ij}} + \ln\left(Var_{ij}\right) \right)$$

$$CCONTR_{2LL,ij} = \left(\frac{Y_{OBSij} - Y_{PREDij}}{SD_{ij}} \right)^2 + 2 \cdot \ln\left(SD_{ij}\right)$$

Dropout Probability

$$CCONTR_{2LL} = -2 \cdot \ln(\Pr)$$

Modified Code

```
$INPUT ID TRT TIME CMT LOCF
DV MDV DVID EVID

$ESTIM MAX=9990 SIG=4 NOABORT
METHOD=CONDITIONAL LAPLACE

$CONTR DATA=(DVID)
$SUBR ADVAN=6 TOL=6
CONTR=contr.for
CCONTR=ccontr_like.for

$MODEL
COMP=(CUMHAZ)
COMP=(HZLAST,INITIALOFF)

$PK
BSHZ=THETA(1) ; Baseline hazard
BETA=THETA(2) ; RD hazard
BET2=THETA(3) ; ID hazard
EFFECT=TRT*THETA(4)
INTRI=(THETA(5)+EFFECT)*EXP(ETA(1))
SLOPI=THETA(6)*EXP(ETA(2))

$DES
DISPRG=INTRI + SLOPI*T
EXPHAZ=EXP(BETA*LOCF + BET2*DISPRG)
DADT(1)=EXPHAZ
DADT(2)=EXPHAZ

$ERROR
CMHZ=BSHZ*A(1) ; Cum hazard overall
HZLA=BSHZ*A(2) ; Cum hazard from last obs
IF (HZLA.LE.0) HZLA=1.0D-10

IF (DVID.EQ.1) THEN
   Y=INTRI + SLOPI*TIME + ERR(1); Status
ENDIF
IF (DVID.EQ.2.AND.DV.EQ.0) THEN
   PD0=EXP(-CMHZ)        ; Pr no dropout
   Y=PD0
ENDIF
IF (DVID.EQ.2.AND.DV.EQ.1) THEN
   PL0=EXP(-(CMHZ-HZLA)) ; Pr no drop last
   PU1=1-EXP(-HZLA)      ; Pr drop unknown
   Y=PL0 * PU1           ; Pr dropout
ENDIF
```

CCONTR

```
      SUBROUTINE CCONTR (ICALL,CNT,P1,P2,IER1,IER2
      SAVE
C LVR and NO should match values in NSIZES
      PARAMETER(LVR=30,NO=50)
      COMMON /ROCM4/ Y(NO),DATA(NO,3)
      DOUBLE PRECISION CNT,P1,P2,Y
      DIMENSION P1(*),P2(LVR,*)
      TYPE=DATA(1,1)
C Value of TYPE is provided as a user defined data item
      IF (TYPE.EQ.1)THEN
C CELS is used for continuous type data
         CALL CELS(CNT,P1,P2,IER1,IER2)
      ELSE
C CLIK is used for LIKE or -2LL
C first argument is 1 for LIKE and 2 for -2LL
         CALL CLIK(1,CNT,P1,P2,IER1,IER2)
      ENDIF
      RETURN
      END
```

LIKE vs -2LL Methods ID Bias and Imprecision

1000 subjects observed at t_i = 0, 25, 50, 75 and 100
100 replications
Slope 1 u/time SD 1 u
Baseline 0.0001
ID hazard 0.065
Average Dropout 53% (95 percentile 50-56%)



	LIKE	Success	79%	7 h 4 min	-2LL	Success	44%	7h 53 min
	Bias	loCI	hiCI	RMSE	Bias	loCI	hiCI	RMSE
Slope	0.01%	-0.14%	0.17%	0.7%	0.1%	-0.11%	0.3%	0.67%
PPVslope	-0.15%	-0.77%	0.46%	2.7%	-0.23	-0.77	0.32	2.64
Baseline	2.3%	-1.3%	5.8%	15.8%	15%	11%	17%	10%
ID hazard	-0.16%	-0.8%	0.5%	2.9%	-2.2%	-2.7%	-1.7%	1.7%

CRD (null) Randomization Test

1000 subjects observed at t_i = 0, 25, 50, 75 and 100
1000 replications; RD and ID: One extra parameter
Bootstrap mean and 95% confidence interval
Slope 1 u/time SD 1 u
Baseline 0.01
ID hazard 0.0
Average Dropout 50% (95 percentile 47-53%)
Type I error rate 5%



Model Comparison	CritOBJ	Low	High	F success
Null: CRD Alternate: RD	3.75	3.27	4.25	95%
Null: CRD Alternate: ID	3.67	3.33	4.17	95%

- Is Missingness Informative?
  - Only one bit of information per subject
  - Dropout model did not influence disease progress estimates
- Can NONMEM get the right answer?
  - LIKE method with CCONTR is OK
  - Direct coding of -2LL is biased
- Can we distinguish CRD, RD and ID?
  - Randomization test shows ΔOBJ is approximately χ^2 distributed